```
$TITLE: M1-5, Structure of a general-equilibrium model
* simple (almost trivial) example of a one-good, one-factor,
* one-consumer economy

PARAMETERS
 LBAR    labor supply (fixed and inelastic)
 ALPHA   productivity parameter  X = ALPHA*L;

LBAR = 100;
ALPHA = 2;


NONNEGATIVE VARIABLES
 P         price of X
 X         quantity of X
 W         wage rate
 INCOME    income from labor supply;


EQUATIONS
 ZPROFIT      zeroprofits in X production
 CMKTCLEAR    commodity (X) market clearing
 LMKTCLEAR    labor market clearing
 CONSINCOME   consumer income balance;
```

```
ZPROFIT..    W/ALPHA =G= P;

CMKTCLEAR..  X =G= INCOME/P;

LMKTCLEAR..  LBAR =G= X/ALPHA;

CONSINCOME.. INCOME =G= W*LBAR;

MODEL GE /ZPROFIT.X, CMKTCLEAR.P, LMKTCLEAR.W, CONSINCOME.INCOME/;

* set some starting values

P.L = 1;
W.L = 1;
X.L = 200;
INCOME.L = 100;

* choose a numeraire

W.FX = 1;

OPTION MCP = PATH;
SOLVE GE USING MCP;

* double labor productivity
```

```
ALPHA = 4;

SOLVE GE USING MCP;

* change numeraire

W.UP = +INF;
W.LO = 0;
P.FX = 1;

ALPHA = 2;

SOLVE GE USING MCP;

* double labor productivity

ALPHA = 4;

SOLVE GE USING MCP;



$ontext
formulated as an NLP
the first theorem of welfare economics says that a competitive
equilibrium is Pareto optimal
```

```
*in some very simple situation, such as with a single consumer
*this means that equilibrium can also be found as the solution to a
*simple NLP: maximizing utility subject to constraints.
$offtext

ALPHA = 2;
LBAR = 100;

VARIABLE
 U;

EQUATIONS
 OBJECTIVE;

OBJECTIVE..  U =E= X**0.5;

MODEL GE_NLP / OBJECTIVE, ZPROFIT, CMKTCLEAR, LMKTCLEAR, CONSINCOME/;

SOLVE GE_NLP USING NLP MAXIMIZING U;
```